
1. [bookmark: _Toc122434485][bookmark: _Toc96965686]Overview
This document lists all the essencial changes needed to correct problems in the TTCN implementation of IMS test case 7.20 which is part of the IMS_NR5GC test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 
Contact:	Pengchong Shen
	shenpengchong@cictmobile.com
Parikshit Bhise
	Parikshit.bhise@rohde-schwarz.com	
0. [bookmark: _Toc96965687]Verification Test Summary
Test Case: 	7.20
ATS Version:	iwd-TTCN3-B2020-09_D21wk49.zip
System Simulator used:	Datang LinkTester 5G Protocol Conformance Test System ECT9610
                                              Rohde & Schwarz 5G Protocol Conformance Test platform
UE used:	HiSilicon Balong 5000 UE, Samsung SM-G996U UE
Verification Status:	PASS

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc96523148][bookmark: _Toc96965688]Corrections required
1. [bookmark: _Toc96523150][bookmark: _Toc96965689][bookmark: _Toc67572159][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665]Change 1
	Function name
	fl_BuildRemotePortRange
f_BuildDefaultQoSRules

	Reason for change
	1. Video RTP/RTCP port shall use 60352/60353.
2. According to Table 4.9.27.3-1: PDU SESSION MODIFICATION COMMAND (step 1, Table 4.9.27.2.2-1)
 of TS38508-1-h30_s046-s04b, 4th parameter of cs_QoS_Rule of branch (p_RuleNum == "7_Video") is wrong because “Operation code” is “001”B.

	Summary of change
	1. Add parameter of p_Port to function fl_BuildRemotePortRange to transfer port number of video.
2. Modify template cs_PacketFilterList_MediaUDP definition, add 2nd parameter of p_Port to transfer port number of video and use tsc_IMS_MediaPort_M2.
3. Modify 4th parameter of cs_QoS_Rule of branch (p_RuleNum == "7_Video") and (p_RuleNum == "7_VoiceVideo") of function f_BuildDefaultQoSRules from '011'B to -, which means using default value of '001'B.
4. Add “import from CommonIP all;” to NG_NAS_Templates.ttcn module.

	TTCN module
	NG_NAS_Common.ttcn
NG_NAS_Templates.ttcn


	MCC160 Comment
	



Before Change
	    function fl_BuildRemotePortRange () return octetstring
  {
    var octetstring v_PacketFilter;
    
    v_PacketFilter := '51'O; // id = remote port range
    v_PacketFilter := v_PacketFilter & int2oct(tsc_IMS_MediaPort_M1, 2) & int2oct(tsc_IMS_MediaPort_M1+1, 2);
    return v_PacketFilter;
  }

function f_BuildDefaultQoSRules (charstring p_RuleNum,
                                   template (omit) IEI8_Type p_IEI := omit) return template (value) QoS_Rules
  {
    var template (value) QoS_Rules v_QoS_Rules;
    var template (value) QoS_Rule v_VoiceQoS := cs_QoS_Rule('03'O, int2bit(7, 6), '01'O, -, cs_PacketFilterList_MediaUDP('0110'B), '0'B); // @sic R5-210627 sic@;
    
    if (p_RuleNum == "7_Voice") {
      v_QoS_Rules := cs_QoS_Rules ({v_VoiceQoS}, p_IEI) // @sic R5-213440 sic@
    } else if (p_RuleNum == "7_Video") {
      v_QoS_Rules := cs_QoS_Rules ({cs_QoS_Rule('04'O, int2bit(8, 6), '02'O, '011'B, cs_PacketFilterList_MediaUDP('0111'B), '0'B)}, p_IEI) // @sic R5-217796 sic@
    } else if (p_RuleNum == "7_VoiceVideo") {
      v_QoS_Rules := cs_QoS_Rules ({v_VoiceQoS, cs_QoS_Rule('04'O, int2bit(8, 6), '02'O, -, cs_PacketFilterList_MediaUDP('0111'B), '0'B)}, p_IEI) // @sic R5-210627, R5-216161 sic@
    } else {
      v_QoS_Rules := cs_QoS_Rules ({f_BuildDefaultQoSRule(p_RuleNum)}, p_IEI);
    }
    return v_QoS_Rules;
  }

  template (value) NG_PacketFilterList cs_PacketFilterList_MediaUDP (B4_Type p_Id) :=
  { /* @status    APPROVED (IMS, NR5GC, NR5GC_IRAT, POS) */
    new := {
      cs_NG_PacketFilter_Type2 ('11'B,       // bi-directional
                                p_Id,
                                fl_BuildRemotePortRange() & fl_BuildProtocolIdNextHeaderUDP())                                                         // Protocol id / Next Header UDP
    }
  };




After Change
	   function fl_BuildRemotePortRange (integer p_Port := tsc_IMS_MediaPort_M1) return octetstring
  {
    var octetstring v_PacketFilter;
    
    v_PacketFilter := '51'O; // id = remote port range
    v_PacketFilter := v_PacketFilter & int2oct(p_Port, 2) & int2oct(p_Port+1, 2);
    return v_PacketFilter;
  }

function f_BuildDefaultQoSRules (charstring p_RuleNum,
                                   template (omit) IEI8_Type p_IEI := omit) return template (value) QoS_Rules
  {
    var template (value) QoS_Rules v_QoS_Rules;
    var template (value) QoS_Rule v_VoiceQoS := cs_QoS_Rule('03'O, int2bit(7, 6), '01'O, -, cs_PacketFilterList_MediaUDP('0110'B), '0'B); // @sic R5-210627 sic@;
    
    if (p_RuleNum == "7_Voice") {
      v_QoS_Rules := cs_QoS_Rules ({v_VoiceQoS}, p_IEI) // @sic R5-213440 sic@
    } else if (p_RuleNum == "7_Video") {
      v_QoS_Rules := cs_QoS_Rules ({cs_QoS_Rule('04'O, int2bit(8, 6), '02'O, '001'B, cs_PacketFilterList_MediaUDP('0111'B, tsc_IMS_MediaPort_M2), '0'B)}, p_IEI) // @sic R5-217796 sic@
    } else if (p_RuleNum == "7_VoiceVideo") {
      v_QoS_Rules := cs_QoS_Rules ({v_VoiceQoS, cs_QoS_Rule('04'O, int2bit(8, 6), '02'O, -, cs_PacketFilterList_MediaUDP('0111'B, tsc_IMS_MediaPort_M2), '0'B)}, p_IEI) // @sic R5-210627, R5-216161 sic@
    } else {
      v_QoS_Rules := cs_QoS_Rules ({f_BuildDefaultQoSRule(p_RuleNum)}, p_IEI);
    }
    return v_QoS_Rules;
  }

  template (value) NG_PacketFilterList cs_PacketFilterList_MediaUDP (B4_Type p_Id, integer p_Port := tsc_IMS_MediaPort_M1) :=
  { /* @status    APPROVED (IMS, NR5GC, NR5GC_IRAT, POS) */
    new := {
      cs_NG_PacketFilter_Type2 ('11'B,       // bi-directional
                                p_Id,
                                fl_BuildRemotePortRange(p_Port) & fl_BuildProtocolIdNextHeaderUDP())                                                         // Protocol id / Next Header UDP
    }
  };



1. [bookmark: _Toc96523151][bookmark: _Toc96965690]Change 2
	Function name
	f_IMS_MOCallSetup_Video_Step2

	Reason for change
	This scenario is re-INVITE so condition A11 which means INVITE for creating a video is not met, and shall be set false.

	Summary of change
	Set condition v_A11 to false before using it.


	TTCN module
	IMS_Procedures_Video.ttcn

	MCC160 Comment
	



Before Change
	  function f_IMS_MOCallSetup_Video_Step2(VideoCallTypeMO_Type p_VideoCallType,
                                         template SipUrl p_OtherCalleeUri := omit,
                                         IMS_SDP_AudioCodecList_Type p_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR,
                                         IMS_SDP_VideoCodecList_Type p_VideoCodecList := tsc_IMS_SDP_VideoCodecList_H264_only,
                                         boolean p_WithPreconditions := true) runs on IMS_PTC return IMS_InviteRequestWithSdp_Type
  { /* @sic R5w140112: return type changed to IMS_InviteRequestWithSdp_Type to carry decoded SDP offer */
    /* @sic R5s150204: p_OtherCalleeUri charstring -> SipUrl sic@ */
    /* @sic R5-211433: p_AudioCodecList, p_VideoCodecList sic@ */
    /* @sic R5-213070: p_WithPreconditions sic@ */
    var template (present) SDP_Message v_SDP_Message_Step2;
    var INVITE_A_2_1_Context_Type v_Context;
    var IMS_DATA_REQ v_IMS_INVITE_REQ;
    var SDP_Message v_SDP_Message;
    var boolean v_A11 := true;                              /* @sic R5-176371: INVITE for creating a video call sic@ */
    var boolean v_CheckForbiddenCodecParameters := f_IMS_PTC_RanTypeIsNR5GC();
    var boolean v_PreconditionsSupported := p_WithPreconditions;        /* @sic R5-214386: no precondition tag needed in Supported header field in A.15.2 sic@ */
    var OptionTag_List v_OptionTagsForSupported := f_IMS_Invite_OptionTagsSupported(v_PreconditionsSupported);
    var IMS_SDP_AudioCodec_Type v_AudioCodec;
    
    select (p_VideoCallType) {
      case (SetupVideoCall) {
        v_SDP_Message_Step2 := f_IMS_BuildSDP_InitialOfferAudioVideo_RX(p_AudioCodecList, p_VideoCodecList, p_WithPreconditions);
        v_Context := A_2_1_A4;
      }
      case (AddVideo) {
        v_AudioCodec := p_AudioCodecList[0];                /* @sic R5-213073 sic@ */
        v_SDP_Message_Step2 := f_IMS_BuildSDP_InitialOfferAddVideo_RX(v_AudioCodec, p_VideoCodecList, p_WithPreconditions);
        v_Context := A_2_1_A5;
      }
    }

    // Step 2. Receive INVITE Request
    v_IMS_INVITE_REQ := f_IMS_INVITE_ReceiveRequest(v_Context, v_OptionTagsForSupported, p_OtherCalleeUri, -, -, v_A11);  /* @sic R5s150180: p_A10 removed sic@ @sic R5-176209: v_A12 removed sic@ */
    v_SDP_Message := f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP(v_IMS_INVITE_REQ.Request.Invite.messageBody, v_SDP_Message_Step2);

    // check media description:
    if (p_VideoCallType == SetupVideoCall) {                                                 /* @sic R5-172931: no check of voice codecs for 17.1 sic@ */
      f_IMS_SDP_CheckMediaDescr_AudioCodecs(v_SDP_Message.media_list[0], p_AudioCodecList, v_CheckForbiddenCodecParameters);    /* @sic R5-151968: f_IMS_SDP_CheckMediaDescr_AMR_8000_1 replaced by f_IMS_SDP_CheckMediaDescr_AudioCodec sic@
                                                                                                                                   @sic R5-168032, R5-168141: AMR_16000 sic@
                                                                                                                                   @sic R5-172931 sic@ */
    }
    /* @sic R5-211433: f_IMS_SDP_CheckMediaDescr_H264_90000 removed as codecs are specified in f_SDP_MediaAttributesVideoCodecListRX sic@ */
    f_IMS_SDP_CheckMediaDescr_TcapPcfg(v_SDP_Message.media_list[1]);

    return f_IMS_InviteRequestWithSdpMO(v_IMS_INVITE_REQ, v_SDP_Message);
  }



After Change
	   function f_IMS_MOCallSetup_Video_Step2(VideoCallTypeMO_Type p_VideoCallType,
                                         template SipUrl p_OtherCalleeUri := omit,
                                         IMS_SDP_AudioCodecList_Type p_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR,
                                         IMS_SDP_VideoCodecList_Type p_VideoCodecList := tsc_IMS_SDP_VideoCodecList_H264_only,
                                         boolean p_WithPreconditions := true) runs on IMS_PTC return IMS_InviteRequestWithSdp_Type
  { /* @sic R5w140112: return type changed to IMS_InviteRequestWithSdp_Type to carry decoded SDP offer */
    /* @sic R5s150204: p_OtherCalleeUri charstring -> SipUrl sic@ */
    /* @sic R5-211433: p_AudioCodecList, p_VideoCodecList sic@ */
    /* @sic R5-213070: p_WithPreconditions sic@ */
    var template (present) SDP_Message v_SDP_Message_Step2;
    var INVITE_A_2_1_Context_Type v_Context;
    var IMS_DATA_REQ v_IMS_INVITE_REQ;
    var SDP_Message v_SDP_Message;
    var boolean v_A11 := true;                              /* @sic R5-176371: INVITE for creating a video call sic@ */
    var boolean v_CheckForbiddenCodecParameters := f_IMS_PTC_RanTypeIsNR5GC();
    var boolean v_PreconditionsSupported := p_WithPreconditions;        /* @sic R5-214386: no precondition tag needed in Supported header field in A.15.2 sic@ */
    var OptionTag_List v_OptionTagsForSupported := f_IMS_Invite_OptionTagsSupported(v_PreconditionsSupported);
    var IMS_SDP_AudioCodec_Type v_AudioCodec;
    
    select (p_VideoCallType) {
      case (SetupVideoCall) {
        v_SDP_Message_Step2 := f_IMS_BuildSDP_InitialOfferAudioVideo_RX(p_AudioCodecList, p_VideoCodecList, p_WithPreconditions);
        v_Context := A_2_1_A4;
      }
      case (AddVideo) {
        v_AudioCodec := p_AudioCodecList[0];                /* @sic R5-213073 sic@ */
        v_SDP_Message_Step2 := f_IMS_BuildSDP_InitialOfferAddVideo_RX(v_AudioCodec, p_VideoCodecList, p_WithPreconditions);
        v_Context := A_2_1_A5;
		v_A11 := false;
      }
    }

    // Step 2. Receive INVITE Request
    v_IMS_INVITE_REQ := f_IMS_INVITE_ReceiveRequest(v_Context, v_OptionTagsForSupported, p_OtherCalleeUri, -, -, v_A11);  /* @sic R5s150180: p_A10 removed sic@ @sic R5-176209: v_A12 removed sic@ */
    v_SDP_Message := f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP(v_IMS_INVITE_REQ.Request.Invite.messageBody, v_SDP_Message_Step2);

    // check media description:
    if (p_VideoCallType == SetupVideoCall) {                                                 /* @sic R5-172931: no check of voice codecs for 17.1 sic@ */
      f_IMS_SDP_CheckMediaDescr_AudioCodecs(v_SDP_Message.media_list[0], p_AudioCodecList, v_CheckForbiddenCodecParameters);    /* @sic R5-151968: f_IMS_SDP_CheckMediaDescr_AMR_8000_1 replaced by f_IMS_SDP_CheckMediaDescr_AudioCodec sic@
                                                                                                                                   @sic R5-168032, R5-168141: AMR_16000 sic@
                                                                                                                                   @sic R5-172931 sic@ */
    }
    /* @sic R5-211433: f_IMS_SDP_CheckMediaDescr_H264_90000 removed as codecs are specified in f_SDP_MediaAttributesVideoCodecListRX sic@ */
    f_IMS_SDP_CheckMediaDescr_TcapPcfg(v_SDP_Message.media_list[1]);

    return f_IMS_InviteRequestWithSdpMO(v_IMS_INVITE_REQ, v_SDP_Message);
  }


1. [bookmark: _Toc96523152][bookmark: _Toc96965691]Change 3
	Function name
	f_NR5GC_ModifyPDUSession_AddVideoToSpeech

	Reason for change
	1. Varible v_EPSParam shall be set to encvalue(f_Get_508_EPSQoS(DED_3)) and function f_Get_508_EPSQoS shall add DED_3 and DED_4branches which causes 2 new templates cds_EPS_QoS_Ded_GBR_QCI_2_tlv and cds_EPS_QoS_Ded_GBR_QCI_1_tlv is defined also.
2. As length of v_DRBList is only 1, variable v_DRBList[1] is invalid.

	Summary of change
	1. Add two new templates: cds_EPS_QoS_Ded_GBR_QCI_2_tlv and cds_EPS_QoS_Ded_GBR_QCI_1_tlv.
2. Modify function f_Get_508_EPSQoS, add DED_3 and DED_4 branches.
3. Set v_NewDRBIdVideo  to v_DRBList[0].

	TTCN module
	NR5GC_IMSProcedures.ttcn
NAS_CommonTemplates.ttcn

	MCC160 Comment
	



Before Change
	  function f_NR5GC_ModifyPDUSession_AddVideoToSpeech(NR_CellId_Type p_NR_CellId) runs on NR5GC_PTC
  { 
…
    if (isvalue(v_GSM_MobilityInfo.EPS_Bearer)) {  // This is only set if the UE supports EPS  @sic R5-216153 sic@
      v_InterworkWithEPS := true;
      v_EPS_BearerIdVideo := int2hex(valueof(f_GetNextEPSBearerId (v_GSM_MobilityInfo, v_GMM_MobilityInfo, IPCAN_VideoCall)), 1); // @sic R5s211486 sic@
      v_EPSParam := encvalue(f_Get_508_EPSQoS(DEF_1));  // FFS: wrong QoS!
//v_EPSParam := encvalue(f_Get_508_EPSQoS(DED_3));   //FFS: prose CR needed!
      v_QoSParamOctetsVideo := bit2oct(v_EPSParam);

      v_EPSParam := encvalue(f_EUTRA_NB_Get_508_TrafficFlowTemplate(3, v_EPS_BearerIdVideo, tsc_PdnType_IPv4v6, PDN_1));
      v_TFTParamOctetsVideo := bit2oct(v_EPSParam);

      v_MappedEPSContextVideo := cs_OneMappedEPSContextTwoParams(hex2oct(v_EPS_BearerIdVideo & '0'H), lengthof(v_QoSParamOctetsVideo) + lengthof(v_TFTParamOctetsVideo), cs_EPSParameter_QoS(v_QoSParamOctetsVideo), cs_EPSParameter_TFT(v_TFTParamOctetsVideo));
      v_MappedEPSBearerContexts := cs_OneMappedEPSBearerContextList (v_MappedEPSContextVideo);
    }
…
    v_DRBList := f_NR_GetNextDRBIds(v_GSM_MobilityInfo, false, 1);
    v_NewDRBIdVideo := v_DRBList[1];
…
}

  /* @desc      Get the Default QoS according to 36.508 clause 4.7.3-6
   * @param     p_BearerContextNumber
   * @return    template (value) EPS_QualityOfService
   * @status    APPROVED (ENDC, IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, NBIOT, NR5GC, NR5GC_IRAT, POS)
   */
  function f_Get_508_EPSQoS(template (omit) BEARER_CONTEXT_TYPE p_BearerContextNumber) return template (value) EPS_QualityOfService
  { /* @sic R5-144797: parameter p_BearerContextNumber sic@ */
    var template (value) EPS_QualityOfService v_QoS;
    var BEARER_CONTEXT_TYPE v_BearerContextNumber := fl_BearerContextNumber_Def(p_BearerContextNumber);
    
    select (v_BearerContextNumber){   /* @sic R5-144797: v_QoS does not depend on pc_IMS anymore sic@ */
      case (DEF_1) {
        v_QoS := cs_508_EPS_QoS_Default_1;        /* Qos acc. to 36.508 Table 6.6.1-1 column 1 */
      }
      case (DEF_3) {                              /* @sic R5-210201, R5s211690 sic@ */
        v_QoS := cds_508_EPS_QoS_Default_3;       /* Qos acc. to 36.508 Table 6.6.1-1 column 3 */
      }
      case else {
        v_QoS := cds_EPS_QoS_Def_nonGBR_QCI;      /* Qos acc. to 36.508 Table 6.6.1-1 column 2 */
      }
    }
    return v_QoS;
  }



After Change
	   function f_NR5GC_ModifyPDUSession_AddVideoToSpeech(NR_CellId_Type p_NR_CellId) runs on NR5GC_PTC
  { 
…
    if (isvalue(v_GSM_MobilityInfo.EPS_Bearer)) {  // This is only set if the UE supports EPS  @sic R5-216153 sic@
      v_InterworkWithEPS := true;
      v_EPS_BearerIdVideo := int2hex(valueof(f_GetNextEPSBearerId (v_GSM_MobilityInfo, v_GMM_MobilityInfo, IPCAN_VideoCall)), 1); // @sic R5s211486 sic@
      v_EPSParam := encvalue(f_Get_508_EPSQoS(DED_3));
      v_QoSParamOctetsVideo := bit2oct(v_EPSParam);

      v_EPSParam := encvalue(f_EUTRA_NB_Get_508_TrafficFlowTemplate(3, v_EPS_BearerIdVideo, tsc_PdnType_IPv4v6, PDN_1));
      v_TFTParamOctetsVideo := bit2oct(v_EPSParam);

      v_MappedEPSContextVideo := cs_OneMappedEPSContextTwoParams(hex2oct(v_EPS_BearerIdVideo & '0'H), lengthof(v_QoSParamOctetsVideo) + lengthof(v_TFTParamOctetsVideo), cs_EPSParameter_QoS(v_QoSParamOctetsVideo), cs_EPSParameter_TFT(v_TFTParamOctetsVideo));
      v_MappedEPSBearerContexts := cs_OneMappedEPSBearerContextList (v_MappedEPSContextVideo);
    }
…
    v_DRBList := f_NR_GetNextDRBIds(v_GSM_MobilityInfo, false, 1);
    v_NewDRBIdVideo := v_DRBList[0];
…
}

  function f_Get_508_EPSQoS(template (omit) BEARER_CONTEXT_TYPE p_BearerContextNumber) return template (value) EPS_QualityOfService
  { /* @sic R5-144797: parameter p_BearerContextNumber sic@ */
    var template (value) EPS_QualityOfService v_QoS;
    var BEARER_CONTEXT_TYPE v_BearerContextNumber := fl_BearerContextNumber_Def(p_BearerContextNumber);
    
    select (v_BearerContextNumber){   /* @sic R5-144797: v_QoS does not depend on pc_IMS anymore sic@ */
      case (DEF_1) {
        v_QoS := cs_508_EPS_QoS_Default_1;        /* Qos acc. to 36.508 Table 6.6.1-1 column 1 */
      }
      case (DEF_3) {                              /* @sic R5-210201, R5s211690 sic@ */
        v_QoS := cds_508_EPS_QoS_Default_3;       /* Qos acc. to 36.508 Table 6.6.1-1 column 3 */
      }  
	  case (DED_3) {	
	  	v_QoS := cds_EPS_QoS_Ded_GBR_QCI_2_tlv;
	  }
case (DED_4) {	
	  	v_QoS := cds_EPS_QoS_Ded_GBR_QCI_1_tlv;
	  }
      case else {
        v_QoS := cds_EPS_QoS_Def_nonGBR_QCI;      /* Qos acc. to 36.508 Table 6.6.1-1 column 2 */
      }
    }
    return v_QoS;
  }

    template (value) EPS_QualityOfService cds_EPS_QoS_Ded_GBR_QCI_1_tlv  :=	  
  { //See reference dedicated EPS bearer context #4 in TS 36.508-h10 table 6.6.2-1
    iei := omit,
    iel := '09'O,
    qci                    := '00000001'B,
    maxBitRateUl           := '01101000'B,                //384 kbps
    maxBitRateDl           := '01101000'B,                //384 kbps
    guaranteedBitRateUl    := '01001000'B,                //128 kbps
    guaranteedBitRateDl    := '01001000'B,                //128 kbps
    maxBitRateUlExt        := '00000000'B,
    maxBitRateDlExt        := '00000000'B,
    guaranteedBitRateUlExt := '00000000'B,
    guaranteedBitRateDlExt := '00000000'B,
    maxBitRateUL_Ext2         := omit,  // @sic R5s1300195 Baseline Moving, R5s180640 sic@
    maxBitRateDL_Ext2       := omit,  // @sic R5s1300195 Baseline Moving sic@
    guaranteedBitRateUL_Ext2  := omit,  // @sic R5s1300195 Baseline Moving sic@
    guaranteedBitRateDL_Ext2  := omit   // @sic R5s1300195 Baseline Moving sic@
  };       

  template (value) EPS_QualityOfService cds_EPS_QoS_Ded_GBR_QCI_2_tlv  := 
  { //See reference dedicated EPS bearer context #3 in TS 36.508-h10 table 6.6.2-1
    iei := omit,
    iel := '09'O,
    qci                    := '00000010'B,
    maxBitRateUl           := '10010111'B,                //2104 kbps
    maxBitRateDl           := '10010111'B,                //2104 kbps
    guaranteedBitRateUl    := '10010111'B,                //2104 kbps
    guaranteedBitRateDl    := '10010111'B,                //2104 kbps
    maxBitRateUlExt        := '00000000'B,
    maxBitRateDlExt        := '00000000'B,
    guaranteedBitRateUlExt := '00000000'B,
    guaranteedBitRateDlExt := '00000000'B,
    maxBitRateUL_Ext2         := omit,  // @sic R5s1300195 Baseline Moving, R5s180640 sic@
    maxBitRateDL_Ext2       := omit,  // @sic R5s1300195 Baseline Moving sic@
    guaranteedBitRateUL_Ext2  := omit,  // @sic R5s1300195 Baseline Moving sic@
    guaranteedBitRateDL_Ext2  := omit   // @sic R5s1300195 Baseline Moving sic@
  };     
  


1. [bookmark: _Toc96965692]Change 4
	Function name
	f_IMS_BuildSDP_AudioVideo_RX

	Reason for change
	During media removal UE MAY choose to only include media line in SD. Current implementation expects every media in SDP to have attributes present.

This can cause UE only including media line in SDP to fail. Detailed information in R5-220245 document.

	Summary of change
	Add the default value “*” for p_SDP_MediaAttributes_Video and p_SDP_PreconditionAttributes_Video.
Add check for v_Attribute_List_Video existence.

	TTCN module
	IMS_SDP_MessagesVideo.ttcn

	MCC160 Comment
	



Before 
<<SKIPPED CODE>>
  function f_IMS_BuildSDP_AudioVideo_RX(template (present) SDP_time p_TimeDescription := cr_SDP_Time_0_0,
                                        template (present) SDP_media_field p_SDP_Media_Field_Audio,
                                        template (present) SDP_attribute_list p_SDP_MediaAttributes_Audio,
                                        template (present) SDP_attribute_list p_SDP_PreconditionAttributes_Audio,
                                        template (present) SDP_media_field p_SDP_Media_Field_Video,
                                        template (present) SDP_attribute_list p_SDP_MediaAttributes_Video,
                                        template (present) SDP_attribute_list p_SDP_PreconditionAttributes_Video) runs on IMS_PTC return template (present) SDP_Message
  {
    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();
    var template (present) charstring v_IPAddrOrigin := ?;
    var template (present) charstring v_IPAddrAudio := ?;
    var template (present) charstring v_IPAddrVideo := ?;
    var template (present) integer v_BandwidthAudioRR := ?;
    var template (present) SDP_attribute_list v_Attribute_List_Audio := f_SDP_Attributes_Concat_RX(p_SDP_MediaAttributes_Audio, p_SDP_PreconditionAttributes_Audio);
    var template (present) SDP_attribute_list v_Attribute_List_Video := f_SDP_Attributes_Concat_RX(p_SDP_MediaAttributes_Video, p_SDP_PreconditionAttributes_Video);
    var template (present) SDP_media_desc v_Media_Desc_Audio;
    var template (present) SDP_media_desc v_Media_Desc_Video;
    var template SDP_bandwidth_list v_BandwidthList_Audio;
    var template SDP_bandwidth_list v_BandwidthList_Video;
    var template (present) SDP_media_desc_list v_Media_Desc_List;

    if (f_IMS_PTC_RanTypeIsNR5GC()) {    // @sic R5-211433 sic@
      v_BandwidthAudioRR := (1..infinity);                                                                              // not specified in 34.229-1 C.25
    }

    v_BandwidthList_Audio := f_SDP_SDP_Media_Desc_BandwidthList(p_SDP_Media_Field_Audio, -, -, v_BandwidthAudioRR);     /* results in * if the audio stream shall be released */
    v_BandwidthList_Video := f_SDP_SDP_Media_Desc_BandwidthList(p_SDP_Media_Field_Video);                               /* results in * if the video stream shall be released */

    v_Attribute_List_Audio := superset(all from v_Attribute_List_Audio);
    v_Attribute_List_Video := superset(all from v_Attribute_List_Video);

    v_Media_Desc_Audio := cr_SDP_Media_Desc(p_SDP_Media_Field_Audio,
                                            cr_SDP_Connection(v_AddrType, v_IPAddrAudio),
                                            v_BandwidthList_Audio,
                                            v_Attribute_List_Audio);
    v_Media_Desc_Video := cr_SDP_Media_Desc(p_SDP_Media_Field_Video,
                                            cr_SDP_Connection(v_AddrType, v_IPAddrVideo),
                                            v_BandwidthList_Video,
                                            v_Attribute_List_Video);
    v_Media_Desc_List := { v_Media_Desc_Audio, v_Media_Desc_Video };
    
    return cr_SDP_Message_Common(-, v_AddrType, v_IPAddrOrigin, ?, -, p_TimeDescription, v_Media_Desc_List);
  }
<<SKIPPED CODE>>

After
<<SKIPPED CODE>>
  function f_IMS_BuildSDP_AudioVideo_RX(template (present) SDP_time p_TimeDescription := cr_SDP_Time_0_0,
                                        template (present) SDP_media_field p_SDP_Media_Field_Audio,
                                        template (present) SDP_attribute_list p_SDP_MediaAttributes_Audio,
                                        template (present) SDP_attribute_list p_SDP_PreconditionAttributes_Audio,
                                        template (present) SDP_media_field p_SDP_Media_Field_Video,
                                        template SDP_attribute_list p_SDP_MediaAttributes_Video := *,
                                        template SDP_attribute_list p_SDP_PreconditionAttributes_Video := *) runs on IMS_PTC return template (present) SDP_Message
  {
    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();
    var template (present) charstring v_IPAddrOrigin := ?;
    var template (present) charstring v_IPAddrAudio := ?;
    var template (present) charstring v_IPAddrVideo := ?;
    var template (present) integer v_BandwidthAudioRR := ?;
    var template (present) SDP_attribute_list v_Attribute_List_Audio := f_SDP_Attributes_Concat_RX(p_SDP_MediaAttributes_Audio, p_SDP_PreconditionAttributes_Audio);
    var template SDP_attribute_list v_Attribute_List_Video := f_SDP_Attributes_Concat_RX(p_SDP_MediaAttributes_Video, p_SDP_PreconditionAttributes_Video);
    var template (present) SDP_media_desc v_Media_Desc_Audio;
    var template (present) SDP_media_desc v_Media_Desc_Video;
    var template SDP_bandwidth_list v_BandwidthList_Audio;
    var template SDP_bandwidth_list v_BandwidthList_Video;
    var template (present) SDP_media_desc_list v_Media_Desc_List;

    if (f_IMS_PTC_RanTypeIsNR5GC()) {    // @sic R5-211433 sic@
      v_BandwidthAudioRR := (1..infinity);                                                                              // not specified in 34.229-1 C.25
    }

    v_BandwidthList_Audio := f_SDP_SDP_Media_Desc_BandwidthList(p_SDP_Media_Field_Audio, -, -, v_BandwidthAudioRR);     /* results in * if the audio stream shall be released */
    v_BandwidthList_Video := f_SDP_SDP_Media_Desc_BandwidthList(p_SDP_Media_Field_Video);                               /* results in * if the video stream shall be released */

    v_Attribute_List_Audio := superset(all from v_Attribute_List_Audio);
    if (ispresent(v_Attribute_List_Video)){   
    v_Attribute_List_Video := superset(all from v_Attribute_List_Video);
    } 
    v_Media_Desc_Audio := cr_SDP_Media_Desc(p_SDP_Media_Field_Audio,
                                            cr_SDP_Connection(v_AddrType, v_IPAddrAudio),
                                            v_BandwidthList_Audio,
                                            v_Attribute_List_Audio);
    v_Media_Desc_Video := cr_SDP_Media_Desc(p_SDP_Media_Field_Video,
                                            cr_SDP_Connection(v_AddrType, v_IPAddrVideo),
                                            v_BandwidthList_Video,
                                            v_Attribute_List_Video);
    v_Media_Desc_List := { v_Media_Desc_Audio, v_Media_Desc_Video };
    
    return cr_SDP_Message_Common(-, v_AddrType, v_IPAddrOrigin, ?, -, p_TimeDescription, v_Media_Desc_List);
  }<<SKIPPED CODE>>

1. [bookmark: _Toc96965693]Change 5
	Function name
	f_IMS_BuildSDP_ReleaseVideoOffer_RX

	Reason for change
	During media removal UE MAY choose to only include media line in SD. Current implementation expects every media in SDP to have attributes present.
This can cause UE only including media line in SDP to fail. Detailed information in R5-220245 document.

	Summary of change
	Modified function f_IMS_BuildSDP_ReleaseVideoOffer_RX to not check offer video attributes in SDP message during video remove.


	TTCN module
	IMS_SDP_MessagesVideo.ttcn

	MCC160 Comment
	



Before 
<<SKIPPED CODE>>
  function f_IMS_BuildSDP_ReleaseVideoOffer_RX(SDP_Message p_PreviousSdpOffer,
                                               IMS_SDP_AudioCodec_Type p_AudioCodec := AMR_16000,
                                               IMS_SDP_VideoCodec_Type p_VideoCodec := H264_90000,
                                               boolean p_WithPreconditions := true) runs on IMS_PTC return template (present) SDP_Message
  { /* @sic R5-213073, R5-213074: p_AudioCodec, p_VideoCodec, p_WithPreconditions sic@ */
    var charstring v_FmtAudio := f_SDP_MediaDescr_GetAudioCodec(p_PreviousSdpOffer.media_list[0], p_AudioCodec);        // @sic R5-213073: NR5GC sic@
    var charstring v_FmtVideo := f_SDP_MediaDescr_GetVideoCodec(p_PreviousSdpOffer.media_list[1], p_VideoCodec);        // @sic R5-213073: NR5GC sic@
    var template (present) SDP_attribute_rtpmap_codec v_RtpmapCodecAudio := f_SDP_AttributeRtpmapAudioRX(p_AudioCodec); // @sic R5-213073: NR5GC sic@
    var template (present) SDP_attribute_rtpmap_codec v_RtpmapCodecVideo := f_SDP_AttributeRtpmapVideoRX(p_VideoCodec); // @sic R5-213073: NR5GC sic@
    var template (present) SDP_attribute_list v_SDP_PreconditionAttributes := {};
    var template (present) SDP_attribute_list v_SDP_MediaAttributes_Audio;
    var template (present) SDP_attribute_list v_SDP_MediaAttributes_Video;

    if (p_WithPreconditions) {  // @sic R5-213073, R5-213074: NR5GC sic@
      v_SDP_PreconditionAttributes := cr_SDP_PreconditionAttributes(c_sendrecv, c_sendrecv, c_mandatory, (c_optional, c_mandatory)); /* @sic R5s160604 change 3 sic@ */
    }

    v_SDP_MediaAttributes_Audio := {
      cr_SDP_Attribute_rtpmap(v_FmtAudio, v_RtpmapCodecAudio),
      cr_SDP_Attribute_fmtp(v_FmtAudio)
    };
    v_SDP_MediaAttributes_Video := {
      cr_SDP_Attribute_rtpmap(v_FmtVideo, v_RtpmapCodecVideo),
      cr_SDP_Attribute_fmtp(v_FmtVideo)
    };

    return f_IMS_BuildSDP_AudioVideo_RX(cr_SDP_Time_Any,
                                        cr_SDP_Media_Audio,
                                        v_SDP_MediaAttributes_Audio,
                                        v_SDP_PreconditionAttributes,
                                        cr_SDP_Media(c_video, 0, c_rtpAvpf, cr_SDP_FmtList_AtLeastOne(v_FmtVideo)),  /* port number == 0 => release video stream
                                                                                                                        @sic R5s160604: RTP/AVPF sic@ */
                                        v_SDP_MediaAttributes_Video,
                                        v_SDP_PreconditionAttributes);
  } <<SKIPPED CODE>>

After
<<SKIPPED CODE>>
  function f_IMS_BuildSDP_ReleaseVideoOffer_RX(SDP_Message p_PreviousSdpOffer,
                                               IMS_SDP_AudioCodec_Type p_AudioCodec := AMR_16000,
                                               IMS_SDP_VideoCodec_Type p_VideoCodec := H264_90000,
                                               boolean p_WithPreconditions := true) runs on IMS_PTC return template (present) SDP_Message
  { /* @sic R5-213073, R5-213074: p_AudioCodec, p_VideoCodec, p_WithPreconditions sic@ */
    var charstring v_FmtAudio := f_SDP_MediaDescr_GetAudioCodec(p_PreviousSdpOffer.media_list[0], p_AudioCodec);        // @sic R5-213073: NR5GC sic@
    var charstring v_FmtVideo := f_SDP_MediaDescr_GetVideoCodec(p_PreviousSdpOffer.media_list[1], p_VideoCodec);        // @sic R5-213073: NR5GC sic@
    var template (present) SDP_attribute_rtpmap_codec v_RtpmapCodecAudio := f_SDP_AttributeRtpmapAudioRX(p_AudioCodec); // @sic R5-213073: NR5GC sic@
    var template (present) SDP_attribute_rtpmap_codec v_RtpmapCodecVideo := f_SDP_AttributeRtpmapVideoRX(p_VideoCodec); // @sic R5-213073: NR5GC sic@
    var template (present) SDP_attribute_list v_SDP_PreconditionAttributes := {};
    var template (present) SDP_attribute_list v_SDP_MediaAttributes_Audio;
    
    if (p_WithPreconditions) {  // @sic R5-213073, R5-213074: NR5GC sic@
      v_SDP_PreconditionAttributes := cr_SDP_PreconditionAttributes(c_sendrecv, c_sendrecv, c_mandatory, (c_optional, c_mandatory)); /* @sic R5s160604 change 3 sic@ */
    }

    v_SDP_MediaAttributes_Audio := {
      cr_SDP_Attribute_rtpmap(v_FmtAudio, v_RtpmapCodecAudio),
      cr_SDP_Attribute_fmtp(v_FmtAudio)
    };

    return f_IMS_BuildSDP_AudioVideo_RX(cr_SDP_Time_Any,
                                        cr_SDP_Media_Audio,
                                        v_SDP_MediaAttributes_Audio,
                                        v_SDP_PreconditionAttributes,
                                        cr_SDP_Media(c_video, 0, c_rtpAvpf, cr_SDP_FmtList_AtLeastOne(v_FmtVideo))
                                        );  /* port number == 0 => release video stream
                                             @sic R5s160604: RTP/AVPF sic@ */
                                        
  } 
<<SKIPPED CODE>>

1. [bookmark: _Toc96965694]Change 6
	Function name
	f_SDP_Attributes_Concat_RX

	Reason for change
	During media removal UE MAY choose to only include media line in SDP. Current implementation expects every media in SDP to have attributes present.

This can cause UE only including media line in SDP to fail. Detailed information in R5-220245 document.

	Summary of change
	
Modified auxiliar f_SDP_Attributes_Concat_RX function to be able to handle empty attribute lists.

	TTCN module
	IMS_SDP_Templates.ttcn

	MCC160 Comment
	



Before 
<<SKIPPED CODE>>
  function f_SDP_Attributes_Concat_RX(template (present) SDP_attribute_list p_SDP_Attribute_List1,
                                      template (present) SDP_attribute_list p_SDP_Attribute_List2) return template (present) SDP_attribute_list
  {
    var template (present) SDP_attribute_list v_SDP_Attribute_List := p_SDP_Attribute_List1;
    var integer i := lengthof(p_SDP_Attribute_List1);
    var integer k;

    for (k:=0; k < lengthof(p_SDP_Attribute_List2); k:=k+1) {
      v_SDP_Attribute_List[i+k] := p_SDP_Attribute_List2[k];
    }
    return v_SDP_Attribute_List;
  } 
<<SKIPPED CODE>>

After
<<SKIPPED CODE>>
  function f_SDP_Attributes_Concat_RX(template SDP_attribute_list p_SDP_Attribute_List1,
                                      template SDP_attribute_list p_SDP_Attribute_List2) return template SDP_attribute_list
  {
    var template SDP_attribute_list v_SDP_Attribute_List := p_SDP_Attribute_List1; 
    var integer i;
    var integer k;

    if (isvalue (p_SDP_Attribute_List2)){  
       i := lengthof(p_SDP_Attribute_List1); 
       for (k:=0; k < lengthof(p_SDP_Attribute_List2); k:=k+1) {
         v_SDP_Attribute_List[i+k] := p_SDP_Attribute_List2[k];
       }
    }
    return v_SDP_Attribute_List;
  } 
<<SKIPPED CODE>>



1. [bookmark: _Toc96523153][bookmark: _Toc96965695]Branches executed
(1) This testcase was executed on NR Band 1, nea2, nia2 with HiSilicon Balong 5000 device using Datang LinkTester 5G Protocol Conformance Test System ECT9610.
(2) This testcase was executed on NR band n41 with Samsung SM-G996U device and on NR Band n78 with Hisilicon Balong 5000 device  using NR ciphering algorithm nea1 and integrity algorithm nia1 on Rohde & Schwarz 5G Protocol Conformance Test platform.
1. [bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666][bookmark: _Toc96523154][bookmark: _Toc96965696]Execution Log Files
[bookmark: _Toc43191966][bookmark: _Toc54877395][bookmark: _Toc96523155][bookmark: _Toc96965697]HiSilicon Balong 5000
(1) The HiSilicon Balong 5000 passed this test case on Datang LinkTester 5G Protocol Conformance Test System ECT9610. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test Case Execution log file:
\ TC_7_20_NR5GC.log
· PIXCIT settings used:
\ Pics(HiSilicon-5G IMS).xml
\ Pixit(TC_7_20_NR5GC).xml
(2) The HiSilicon Balong5000 IMS UE passed this test case on Rohde & Schwarz 5G Protocol Conformance Test platform. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
TC_7.20_Hisilicon.log
In the log file (in .html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.
[bookmark: _Toc96451381][bookmark: _Toc96965698]Samsung SM-G996U
The Samsung SM-G996U UE passed this test case on Rohde & Schwarz 5G Protocol Conformance Test platform. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
TC_7.20_Samsung.log
In the log file (in .html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

1. [bookmark: _Toc122434496][bookmark: _Toc295288973][bookmark: _Toc325725668][bookmark: _Toc96523156][bookmark: _Toc96965699]References

